
EC 362 Problem Set #1 
 

1) Using the following MIPS program, determine the instruction format for each instruction and the HEX 
values of each instruction field.  Fill in the table below with those values: 

 
 addi $v0, $zero, 0 # count = 0 
 loop: lw $v1, 0($a0) # load word 
 addi $v0, $v0, 1 # count++ 
 sw $v1, 0($a1) # store word 
 addi $a0, $a0, 4 # ptr to next src 
 addi $a1, $a1, 4 # ptr to next dest 
 bne $v1, $zero, loop # loop if copied word is not 0 
 addi $v0, $v0, -1 # count-- 
 

Instr Opcode Rs Rt Addr/immed 
Addi     
Lw     

Addi     
Sw     

Addi     
Addi     
Bne     
Addi     

     
 
2) Consider the following Java code segment:  
 

 while (save[i] == k) { 
  i = i + 1; 
 } 

 
An inefficient Java translation of this into MIPS might result in assembly code that uses both a 
conditional branch (for the conditional test) and an unconditional jump (for the repeat of the while) 
each time through the loop, in addition to the array bounds checking tests required by Java. For 
example, if the base address of the array save is in register $s6, i is stored in register $s3, and k is 
stored in register $s5, then one possible inefficient MIPS code sequence would be 

 
  lw $t2, 4($s6)  this is the length of "save" 
 loop: slt $t1, $s3, $zero  1 means i < 0 
  bne $t1, $zero, except index out of bounds 
  slt $t1, $s3, $t2  1 means i < save.length 
  beq $t1, $zero, except index out of bounds 
  sll $t1, $s3, 2  4*i 
  add $t1, $t1, $s6  &save[i] - 8 
  lw $t0, 8($t1)  load save[i] 
  bne $t0, $s5, exit  save[i] != k? 
  addi $s3, $s3, 1  i++ 
  j loop 
 exit: 
 

Note that Java Arrays store the length of the array in position $s6 + 4 and then the array itself starts 
at position $s6 + 8.  
 



Only poor compilers would produce code with such excessive loop overhead (3 branches + 1 jump). 
Write a version of the assembly code for this segment so that it uses at most two branches/jumps (total) 
each time through the loop. Please note that regardless of the optimization you perform, Java still 
requires that the array index be "bounds checked" for 0 ≤ i < save.length in every iteration! How many 
instructions are executed before and after the optimization if the number of loop iterations is 100. 

 
3) Consider the following fragment of C code: 
 

 for (i = 0; i <= 100; i++) { 
  a[i] = b[i] + c; 
 } 

 
Assume that a and b are arrays of words and the base address of a is in $a0 and the base address of b 
is in $a1. Register $t0 is associated with variable i and register $s0 with variable c. Write efficient 
MIPS code for this segment. How many instructions are executed during the running of this code? 

 
4) Write a MIPS assembly language function to implement the C / C++ atoi (ASCII to integer) 

function.  The function prototype for atoi is as follows: 
 

int atoi(const char *p); 
 

atoi converts an ASCII decimal string pointed to by the variable p into its corresponding integer 
value. For example, atoi("24") would pass a pointer to the three byte sequence 50 (the ASCII 
decimal value for '2'), 52 (the ASCII decimal value for '4'), 0 (the ASCII decimal value for '\0' which is 
the string termination value) which represents '2', '4', '\0'.  The result would then be the integer 24 
(decimal).  The program need not handle negative numbers, but if any non-digit character appears in 
the string, your program should exit with a return value of -1.  You may use the MUL pseudo-
instruction if needed. Use spim to test your function! 

 



Solutions to Problem Set #1 
 
1)  

Instr Opcode Rs Rt Addr/immed 
Addi 8 0 2 0 
Lw 23 4 3 0 

Addi 8 2 2 1 
Sw 2B 5 3 0 

Addi 8 4 4 4 
Addi 8 5 5 4 
Bne 5 3 0 FFFA 
Addi 8 2 2 FFFF 

     
 
 The address in the bne instruction is a word offset of -6 (FFFA in HEX) since we need to jump 

backwards 6 words (since it is relative to the instruction following the bne).  Note, the book often 
stores this offset field in bytes, which would then give a result of -24 bytes.  But the actual MIPS 
address would be -6 since all instructions must begin on word boundaries. 

 
2)  That unoptimized code results in 1 + 100 * 10 + 8 = 1009 instructions for a 100 iteration loop. 

 
The key observation is that there are only three bounds checking cases that are possible: 
 

i < 0? i < save.length bounds OK? 
1 1 No 
0 1 Yes 
0 0 No 

 
Here is the optimized version with only a single branch or jump (one for the bounds checking and 
another one for the loop test) in each iteration of the loop: 

 
  lw $t2, 4($s6) # this is the length of "save" 
 loop: slt $t0, $s3, $zero # 1 means i < 0 
  slt $t1, $s3, $t2 # 1 means i < save.length 
  beq $t0, $t1, except # index out of bounds 
  sll $t1, $s3, 2 # 4*i 
  add $t1, $t1, $s6 # &save[i] - 8 
  lw $t0, 8($t1) # load save[i] 
  addi $s3, $s3, 1 # i++ (gamble that I need this) 
  beq $t0, $s5, loop # save[i] == k? 
 exit: addi $s3, $s3, -1 # i-- to account for my gamble 
 

This optimized code results in 1 + 101 * 8 + 1 = 810 instructions for a 100 iteration loop. 
 
3)  addi $t6, $zero, 101 # the loop termination value 
  add $t0, $zero, $zero # i = 0 
  addi $t2, $a0, 0 # ptr to current A[i] 
  addi $t3, $a1, 0 # ptr to current B[i] 
 loop: lw $t4, 0($t3) # load B[i] 
  add $t4, $t4, $s0 # B[i] + c 
  sw $t4, 0($t2) # store in A[i] 
  addi $t0, $t0, 1 # i++ 
  addi $t2, $t2, 4 # ptr to next A[i] 
  addi $t3, $t3, 4 # ptr to next B[i] 



  bne $t0, $t6, loop # if i < 101, goto loop 
 

The number of instructions executed is 4 + 7 * 101 = 711.  Note: I tried to be a "smart compiler".  
I know that the loop runs at least once (since 0 <= 100) and I also know that "less than or equal to 
100" is the same as "less than 101" when the loop index is an integer, and I saved this value in a 
register so I could do the test and branch at the same time. 

  
4) atoi in MIPS 
 

atoi: addi $t0, $a0, 0 # copy ptr "p" into $t0 
 addi $v0, $zero, 0 # clear $v0 (the result) 
 addi $t3, $zero, 10 # put 10 into $t3 (for mult later) 
loop: lb $t1, 0($t0) # load current byte (char) 
 beq $t1, $zero, done # test for '\0' 
 addi $t1, $t1, -48 # adjust ASCII values '0' -> 0 
 slti $t2, $t1, 0 # is value < 0? 
 bne $t2, $zero, fault # yes, prepare to exit 
 slti $t2, $t1, 10 # is value < 10? 
 beq $t2, $zero, fault # no, prepare to exit 
 mul $v0, $v0, $t3 # value *= 10 (this is a pseudoinst) 
 add $v0, $v0, $t1 # value += digit 
 addi $t0, $t0, 1 # ptr++ 
 j loop # try next next byte 
fault: addi $v0, $zero, -1 # fault value 
done: jr $ra # return 

